Appendix V

DOE-2 Input File for Base Case Office Building

```
$=========================================================================
$ TYPICAL BUILDINGS FOR BUILDING ENERGY SIMULATION IN HONG KONG
$ By  Sam C M Hui                    Last Update:  1 Jul 1995
$ Base Case Office Building (BCOFFICE.INP)
$=========================================================================
TITLE    LINE-1 *BASE CASE OFFICE BUILDING*
         LINE-2 *DOE-2.1EX SIMULATION RUN *
         LINE-3 *WTH NAME = WEATHR.TMP*
         LINE-4 *BDL NAME = BCOFFICE.INP*
         LINE-5 *BY  SAM C M HUI*  ..

$-------------------------------------------------------------------
$ LOADS INPUT
$-------------------------------------------------------------------
INPUT LOADS
         INPUT-UNITS     METRIC
         OUTPUT-UNITS    METRIC  ..    $—Specify metric units—

ABORT          ERRORS  ..
DIAGNOSTIC     WARNINGS  ..
RUN-PERIOD     JAN 1 1997 THRU DEC 31 1997  ..

B-L  ALT=33  G-A=49000  HOL=NO  D-S=NO  AZ=0  ..
L-R  V=(LV-B,LV-D,LV-F,LV-I,LV-K)  S=(LS-A,LS-C,LS-D,LS-E,LS-F)  ..

$—LOADS BASIC PARAMETERS—
PARAMETER
    SC=.4             $ Shading coefficient for windows
    WWR=.441176       $ Window-to-wall ratio
    SPC-TEMP=25.5     $ Space temperature for load calculation (deg C)
    EQUIP-P=15        $ Equipment load (W/sq.m)
    LTG-P=20          $ Lighting power (W/sq.m)
    OCCUP-P=5         $ Occupancy density (sq.m/psn)
    W-ABSORP=.7       $ Solar absorptance for wall
    R-ABSORP=.7       $ Solar absorptance for roof
    F-ABSORP=.7       $ Solar absorptance for floor
    GC=7.96           $ Glass conductance for windows
    INFIL=.6          $ Infiltration rate (ACH)
    FM=38             $ Floor mulltiplier for middle typical floors
    ..

$—OCCUPANCY SCHEDULE—
OCC-WKS   =D-SCH    (1,7)VALUES=(0) (8,9)(.1,.2) (10,12)(.95) (13)(.5)
              (14,17)(.95) (18)(.3) (19,22)(.1) (23,24)(.05)  ..
OCC-SAT   =D-SCH    (1,7)VALUES=(0) (8,9)(.1,.2) (10,13)(.9) (14,17)(.1)
              (18,19)(.05) (20,24)(0)  ..
OCC-SUN   =D-SCH    (1,7)VALUES=(0) (8,18)(.05) (19,24)(0)  ..
OCC-DAY   =W-SCH    (WD) OCC-WKS  (SAT) OCC-SAT  (SUN,HOL) OCC-SUN  ..
OCC-SCH   =SCH      THRU DEC 31 OCC-DAY  ..

$—LIGHTING SCHEDULE—
LGP-WKS   =D-SCH    (1,6)VALUES=(.05) (7,8)(.1) (9)(.3) (10,12)(.9) (13)(.8)
              (14,17)(.9) (18,24)(.5,.3,.3,.2,.2,.1,.05)  ..
LGP-SAT   =D-SCH    (1,6)VALUES=(.05) (7,8)(.1) (9)(.3) (10,13)(.9)
              (14,17)(.15) (18,24)(.05)  ..
LGP-SUN   =D-SCH    (1,7)VALUES=(.05) (8,18)(.1) (19,24)(.05)  ..
LGP-DAY   =W-SCH    (WD) LGP-WKS  (SAT) LGP-SAT  (SUN,HOL) LGP-SUN  ..
LGP-SCH   =SCH      THRU DEC 31 LGP-DAY  ..
```

```
$—OFFICE EQUIPMENT SCHEDULE—
EQP-WKS  =D-SCH   (1,9)VALUES=(.02) (10,12)(.9) (13)(.5) (14,17)(.9)
            (18,24)(.02)  ..
EQP-SAT  =D-SCH   (1,9)VALUES=(.02) (10,13)(.9) (14,17)(.05) (18,24)(.02)  ..
EQP-SUN  =D-SCH   (1,24)VALUES=(.02)  ..
EQP-DAY  =W-SCH   (WD) EQP-WKS  (SAT) EQP-SAT  (SUN,HOL) EQP-SUN  ..
EQP-SCH  =SCH     THRU DEC 31 EQP-DAY  ..

$—INFILTRATION SCHEDULE—
INF-WKS  =D-SCH   (1,8)VALUES=(1) (9,18)(0) (19,24)(1)  ..
INF-SAT  =D-SCH   (1,8)VALUES=(1) (9,13)(0) (14,24)(1)  ..
INF-SUN  =D-SCH   (1,24)VALUES=(1)  ..
INF-DAY  =W-SCH   (WD) INF-WKS  (SAT) INF-SAT  (SUN,HOL) INF-SUN  ..
INF-SCH  =SCH     THRU DEC 31 INF-DAY  ..

$—MATERIALS (DOE-2 Code)—
AIRGAP   =MAT     RES=.1568  ..          $ Air gap (AL21)
AIRVOID  =MAT     RES=.162  ..           $ Air void (AL33)
BUILDUP  =MAT     TH=.0095  COND=.162    $ 10mm felt & membrane
            DENS=1121  S-H=1464  ..  $ roof build-up (BR01)
CARPET   =MAT     RES=.367  ..           $ Carpet on floor (CP01)
FCPANEL  =MAT     TH=.0191  COND=.057    $ Acoustic false ceiling
            DENS=288  S-H=1339  ..  $ panel (AC03)
GLASS    =MAT     RES=.0057  ..          $ 6mm clear glass
GYPBORD  =MAT     TH=.0159  COND=.1602   $ 16mm gypsum board (GP02)
            DENS=801  S-H=837  ..
HWCONC   =MAT     TH=.2032  COND=1.31    $ 200mm H.W. conc. (CC05)
            DENS=2243  S-H=837  ..
LWCONC-1 =MAT     TH=.1524  COND=.3603   $ 150mm L.W. conc. (CC25)
            DENS=1281  S-H=837  ..
LWCONC-2 =MAT     TH=.0762  COND=.3603   $ Half of 150mm L.W. conc.
            DENS=1281  S-H=837  ..  $ for mid-floor & ceiling
MOSAIC   =MAT     TH=.005   COND=1.8016  $ 5mm mosaic tiles (TZ01)
            DENS=2243  S-H=335  ..
MOTAR    =MAT     TH=.0191  COND=.7207   $ 19mm cement motar (CM01)
            DENS=1858  S-H=837  ..
PLASTER  =MAT     TH=.0191  COND=.23     $ 19mm gyp. plaster (GP04)
            DENS=721  S-H=837  ..
PLYWOOD  =MAT     TH=.0191  COND=.1154   $ 19mm plywood board (PW05)
            DENS=545  S-H=1213  ..
RFINSUL  =MAT     TH=.0508  COND=.052    $ 50mm roof insuln. (IN74)
            DENS=256  S-H=837  ..
SCREED   =MAT     TH=.0508  COND=.36     $ 50mm floor screeding (CC23)
            DENS=1281  S-H=837  ..
SLAG     =MAT     TH=.0127  COND=1.44    $ 13mm slag on roof (RG01)
            DENS=881  S-H=1674  ..
WALLPAPER =MAT    RES=.0106  ..          $ Wall paper (BP01)

$—CONSTRUCTIONS—
LAY-EW   =LAYERS  =MAT(GLASS,AIRGAP,PLYWOOD,WALLPAPER)  ..
LAY-RF   =LAYERS  =MAT(SLAG,BUILDUP,RFINSUL,HWCONC,AIRVOID,FCPANEL)  ..
LAY-GF   =LAYERS  =MAT(CARPET,SCREED,HWCONC)  ..
LAY-FLR  =LAYERS  =MAT(CARPET,SCREED,LWCONC-2)  ..
LAY-CLG  =LAYERS  =MAT(LWCONC-2,AIRVOID,FCPANEL)  ..
LAY-INP  =LAYERS  =MAT(GYPBORD,AIRGAP,GYPBORD)  ..
LAY-CWL  =LAYERS  =MAT(MOSAIC,MOTAR,HWCONC,PLASTER,WALLPAPER)  ..

WALL-1   =CONS    LAYERS=LAY-EW  ABS=W-ABSORP  ..
RF       =CONS    LAYERS=LAY-RF  ABS=R-ABSORP  ..
GND-FLR  =CONS    LAYERS=LAY-GF  ..
FLOOR    =CONS    LAYERS=LAY-FLR  ABS=F-ABSORP  ..  $ Half floor $
CEILING  =CONS    LAYERS=LAY-CLG  ..             $ Half Ceiling $
INT-PART =CONS    LAYERS=LAY-INP  ..
COREWALL =CONS    LAYERS=LAY-CWL  ..

$—GLAZING—
GLAZ-1   =GLASS-TYPE   P=1  S-C=SC  G-C=GC  ..

$—SPACE DESCRIPTION—
OFFICE   =SPACE-CONDITIONS   Z-TYPE=CONDITIONED  T=(SPC-TEMP)
                  F-W=342
                  F-TYPE=HEAVY  F-F=0.5  F-WGT=195

                  PEOPLE-HG-LAT=63  PEOPLE-HG-SENS=69
```

```
                    AREA/PERSON=OCCUP-P
                    LIGHTING-W/AREA=LTG-P
                    L-T=REC-FLUOR-NV
                      L-T-S=1.0  L-T-O=0.0  L-T-R=0.0
                      L-R-F=(0.67,0.09)
                    EQUIPMENT-W/AREA=EQUIP-P
                    E-SCH=EQP-SCH  L-SCH=LGP-SCH  P-SCH=OCC-SCH

                    INF-METHOD=AIR-CHANGE
                    A-C=INFIL  I-SCH=INF-SCH  ..

SET FOR   E-W      H=3.4  W=35  CONS=WALL-1  AZ=180  G-R=0.2  ..
SET FOR   WINDOW   H=WWR TIMES 3.4  W=35  X=0  GLASS-TYPE=GLAZ-1
             W-S-T=MOVABLE-INTERIOR  ..
SET FOR   I-W      CONS=INT-PART  I-W-TYPE=STANDARD  ..
SET FOR   SPACE    F-M=FM  ..

$—TYPICAL MIDDLE FLOOR—
$ EW  = exterior wall          WIN = window
$ IW  = interior partition     CW  = corewall

SP-N     =SPACE   S-C=OFFICE  V=341  A=139  X=35  Y=35  AZ=180  ..
   EW-N     =E-W  ..
   WIN-N    =WI  ..
   IW-NW    =I-W     A=15.83  N-T=SP-W  ..
   IW-NE    =I-W     A=15.83  N-T=SP-E  ..
   IW-NI    =I-W     A=63.36  N-T=SP-I  ..
   CL-N     =I-W     I-W-TYPE=ADIABATIC  A=139  TILT=0  CONS=CEILING  ..
   FL-N     =I-W     I-W-TYPE=ADIABATIC  A=139  TILT=180  CONS=FLOOR  ..

SP-E     =SPACE   S-C=OFFICE  V=341  A=139  X=35  Y=0  AZ=-90  ..
   EW-E     =E-W  ..
   WIN-E    =WI  ..
   IW-EI    =I-W     A=63.36  N-T=SP-I  ..
   CL-E     =I-W     LIKE CL-N  A=139  ..
   FL-E     =I-W     LIKE FL-N  A=139  ..

SP-S     =SPACE   S-C=OFFICE  V=341  A=139  X=0  Y=0  AZ=0  ..
   EW-S     =E-W  ..
   WIN-S    =WI  ..
   IW-SE    =I-W     A=15.83  N-T=SP-E  ..
   IW-SW    =I-W     A=15.83  N-T=SP-W  ..
   IW-SI    =I-W     A=63.36  N-T=SP-I  ..
   CL-S     =I-W     LIKE CL-N  A=139  ..
   FL-S     =I-W     LIKE FL-N  A=139  ..

SP-W     =SPACE   S-C=OFFICE  V=341  A=139  X=0  Y=35  AZ=90  ..
   EW-W     =E-W  ..
   WIN-W    =WI  ..
   IW-WI    =I-W     A=63.36  N-T=SP-I  ..
   CL-W     =I-W     LIKE CL-N  A=139  ..
   FL-W     =I-W     LIKE FL-N  A=139  ..

SP-I     =SPACE   S-C=OFFICE  V=1159  A=473  L-SCH=LGP-SCH  ..
   CL-I     =I-W     LIKE CL-N  A=473  ..
   CW-I     =I-W     H=3.4  W=56  N-T=CORE  ..
   FL-I     =I-W     LIKE FL-N  A=473  ..

CORE     =SPACE   Z-TYPE=UNCONDITIONED  V=666.4  A=196
            T=(27.5)  F-W=342  ..

$—TOPMOST FLOOR (WITH ROOF)—
RFSP-N   =SPACE   LIKE SP-N  F-M=1  ..
   RF-N     =ROOF    CONS=RF  TILT=0  H=30.42  W=4.57  ..
   EWR-N    =E-W     LIKE EW-N  ..
   WINR-N   =WI      LIKE WIN-N  ..
   IWR-NW   =I-W     LIKE IW-NW  N-T=RFSP-W  ..
   IWR-NE   =I-W     LIKE IW-NE  N-T=RFSP-E  ..
   IWR-NI   =I-W     LIKE IW-NI  N-T=RFSP-I  ..
   FLR-N    =I-W     LIKE FL-N  ..

RFSP-E   =SPACE   LIKE SP-E  F-M=1  ..
   RF-E     =ROOF    LIKE RF-N  H=30.42  W=4.57  ..
   EWR-E    =E-W     LIKE EW-E  ..
```

```
   WINR-E    =WI      LIKE WIN-E  ..
   IWR-EI    =I-W     LIKE IW-EI  N-T=RFSP-I  ..
   FLR-E     =I-W     LIKE FL-E  ..

RFSP-S    =SPACE    LIKE SP-S  F-M=1  ..
   RF-S      =ROOF     LIKE RF-N  H=30.42  W=4.57  ..
   EWR-S     =E-W      LIKE EW-S  ..
   WINR-S    =WI       LIKE WIN-S  ..
   IWR-SE    =I-W      LIKE IW-SE  N-T=RFSP-E  ..
   IWR-SW    =I-W      LIKE IW-SW  N-T=RFSP-W  ..
   IWR-SI    =I-W      LIKE IW-SI  N-T=RFSP-I  ..
   FLR-S     =I-W      LIKE FL-S  ..

RFSP-W    =SPACE    LIKE SP-W  F-M=1  ..
   RF-W      =ROOF     LIKE RF-N  H=30.42  W=4.57  ..
   EWR-W     =E-W      LIKE EW-W  ..
   WINR-W    =WI       LIKE WIN-W  ..
   IWR-WI    =I-W      A=63.36  N-T=RFSP-I  ..
   FLR-W     =I-W      LIKE FL-W  ..

RFSP-I    =SPACE    LIKE SP-I  F-M=1  L-SCH=LGP-SCH  ..
   RF-I      =ROOF     LIKE RF-N  H=21.75  W=21.75  ..
   CWR-I     =I-W      LIKE CW-I  N-T=RFCORE  ..
   FLR-I     =I-W      LIKE FL-I  ..

RFCORE    =SPACE    LIKE CORE  F-M=1  ..
   RF-CORE   =ROOF     LIKE RF-N  H=14  W=14  ..

$—GROUND FLOOR—
GFSP-N    =SPACE    LIKE SP-N  F-M=1  ..
   EWG-N     =E-W      LIKE EW-N  ..
   WING-N    =WI       LIKE WIN-N  ..
   IWG-NW    =I-W      A=15.83  N-T=GFSP-W  ..
   IWG-NE    =I-W      A=15.83  N-T=GFSP-E  ..
   IWG-NI    =I-W      A=63.36  N-T=GFSP-I  ..
   CLG-N     =I-W      LIKE CL-N  ..
   FLG-N     =U-F      CONS=GND-FLR  TILT=180  A=139  ..

GFSP-E    =SPACE    LIKE SP-E  F-M=1  ..
   EWG-E     =E-W      LIKE EW-E  ..
   WING-E    =WI       LIKE WIN-E  ..
   IWG-EI    =I-W      A=63.36  N-T=GFSP-I  ..
   CLG-E     =I-W      LIKE CL-E  ..
   FLG-E     =U-F      LIKE FLG-N  A=139  ..

GFSP-S    =SPACE    LIKE SP-S  F-M=1  ..
   EWG-S     =E-W      LIKE EW-S  ..
   WING-S    =WI       LIKE WIN-S  ..
   IWG-SE    =I-W      A=15.83  N-T=GFSP-E  ..
   IWG-SW    =I-W      A=15.83  N-T=GFSP-W  ..
   IWG-SI    =I-W      A=63.36  N-T=GFSP-I  ..
   CLG-S     =I-W      LIKE CL-S  ..
   FLG-S     =U-F      LIKE FLG-N  A=139  ..

GFSP-W    =SPACE    LIKE SP-W  F-M=1  ..
   EWG-W     =E-W      LIKE EW-W  ..
   WING-W    =WI       LIKE WIN-W  ..
   IWG-WI    =I-W      A=63.36  N-T=GFSP-I  ..
   CLG-W     =I-W      LIKE CL-W  ..
   FLG-W     =U-F      LIKE FLG-N  A=139  ..

GFSP-I    =SPACE    LIKE SP-I  F-M=1  ..
   CLG-I     =I-W      LIKE CL-I  ..
   CWG-I     =I-W      LIKE CW-I  N-T=GFCORE  ..
   FLG-I     =U-F      LIKE FLG-N  A=473  ..

GFCORE    =SPACE    LIKE CORE  F-M=1  ..
   FL-CORE   =U-F      LIKE FLG-N  A=196  ..

END  ..
COMPUTE LOADS  ..

$———————————————————————————————————————--
$ SYSTEMS INPUT
```

```
$————————————————————————————————————————--
INPUT SYSTEMS
       INPUT-UNITS     METRIC
       OUTPUT-UNITS    METRIC  ..    $—Specify metric units—

S-R  V=(ALL-VERIFICATION)
   S=(SS-A,SS-B,SS-C,SS-D,SS-E,SS-H,SS-I,SS-J,SS-K,SS-M,SS-N)  ..

$—SYSTEMS BASIC PARAMETERS—
PARAMETER
   OA-PERSON=25.2     $ Outdoor air requirements (m3/h/person)
   T-COOL=25.5        $ Cooling design indoor temp (deg C)
   FAN-EFF=.55        $ Supply fan efficiency
   FAN-STATIC=139.7   $ Supply fan static (mm Aq)
   SYS-TYPE=VAVS      $ System type
   T-HEAT=21          $ Heating design indoor temp (deg C)
   COOL-SB=37         $ Cooling setback temp (deg C)
   HEAT-SB=10         $ Heating setback temp (deg C)
   THR-RNG=1.1        $ Throttling range (deg C)
   ..

$—SYSTEMS SCHEDULES—
FAN-WKS   =D-SCH    (1,8)VALUES=(0) (9,18) (1) (19,24) (0)  ..
FAN-SAT   =D-SCH    (1,8)VALUES=(0) (9,13) (1) (14,24) (0)  ..
FAN-SUN   =D-SCH    (1,24)VALUES=(0)  ..
FAN-DAY   =W-SCH    (WD) FAN-WKS  (SAT) FAN-SAT  (SUN,HOL) FAN-SUN  ..
FAN-SCH   =SCH      THRU DEC 31 FAN-DAY  ..

COOL-WKS  =D-SCH    (1,8)TEMP=(COOL-SB) (9,18) (T-COOL) (19,24) (COOL-SB)  ..
COOL-SAT  =D-SCH    (1,8)TEMP=(COOL-SB) (9,13) (T-COOL) (14,24) (COOL-SB)  ..
COOL-SUN  =D-SCH    (1,24)TEMP=(COOL-SB)  ..
COOL-DAY  =W-SCH    (WD) COOL-WKS  (SAT) COOL-SAT  (SUN,HOL) COOL-SUN  ..
COOL-SCH  =SCH      THRU DEC 31 COOL-DAY  ..

HEAT-WKS  =D-SCH    (1,8)TEMP=(HEAT-SB) (9,18) (T-HEAT) (19,24) (HEAT-SB)  ..
HEAT-SAT  =D-SCH    (1,8)TEMP=(HEAT-SB) (9,13) (T-HEAT) (14,24) (HEAT-SB)  ..
HEAT-SUN  =D-SCH    (1,24)TEMP=(HEAT-SB)  ..
HEAT-DAY  =W-SCH    (WD) HEAT-WKS  (SAT) HEAT-SAT  (SUN,HOL) HEAT-SUN  ..
HEAT-SCH  =SCH      THRU DEC 31 HEAT-DAY  ..

COOL-OFF  =SCHEDULE THRU DEC 31 (ALL) (1,24)VALUES=(1)  ..
HEAT-OFF  =SCHEDULE THRU DEC 31 (ALL) (1,24)VALUES=(1)  ..

R1       =D-R-SCH  SUPPLY-HI=17.2  SUPPLY-LO=12.2
              OUTSIDE-LO=10   OUTSIDE-HI=30  ..
SAT-RESET =R-SCH    THRU DEC 31 (ALL) R1  ..

$—ZONE & SYSTEM DESCRIPTION—
SET FOR   ZONE      Z-TYPE=CONDITIONED  S-O=FROM-LOADS
              D-H-T=T-HEAT  D-C-T=T-COOL
              H-T-SCH=HEAT-SCH  C-T-SCH=COOL-SCH
              T-TYPE=REVERSE-ACTION  T-R=THR-RNG
              OA-FLOW/PER=OA-PERSON  ..
SP-N      =ZONE  ..
SP-E      =ZONE  ..
SP-S      =ZONE  ..
SP-W      =ZONE  ..
SP-I      =ZONE  ..
CORE      =ZONE     Z-TYPE=UNCONDITIONED  ..
RFSP-N    =ZONE  ..
RFSP-E    =ZONE  ..
RFSP-S    =ZONE  ..
RFSP-W    =ZONE  ..
RFSP-I    =ZONE  ..
RFCORE    =ZONE     LIKE CORE  ..
GFSP-N    =ZONE  ..
GFSP-E    =ZONE  ..
GFSP-S    =ZONE  ..
GFSP-W    =ZONE  ..
GFSP-I    =ZONE  ..
GFCORE    =ZONE     LIKE CORE  ..

SYS-CTRL  =SYSTEM-CONTROL     C-SCH=COOL-OFF  H-SCH=HEAT-OFF  HEAT-SET-T=17.7
                   COOL-CONTROL=RESET  C-R-SCH=SAT-RESET
```

```
                    MAX-H=60  MIN-H=40  MIN-SUPPLY-T=12.2  ..

SYS-AIR   =SYSTEM-AIR        O-CTRL=TEMP  ..

SYS-FAN   =SYSTEM-FANS       F-SCH=FAN-SCH  F-C=INLET  S-S=FAN-STATIC
                    S-E=FAN-EFF  M-P=IN-AIRFLOW  F-P=DRAW-THROUGH
                    N-C-C=STAY-OFF  ..

SYS-TERM  =SYSTEM-TERMINAL    REHEAT-DELTA-T=5  MIN-CFM-RATIO=0.3  ..

SYST-1    =SYSTEM   S-TYPE=SYS-TYPE  S-C=SYS-CTRL  S-A=SYS-AIR
               S-FANS=SYS-FAN  S-T=SYS-TERM  R-A-P=DUCT
               Z-N=(SP-N,SP-E,SP-S,SP-W,SP-I,CORE,
                   RFSP-N,RFSP-E,RFSP-S,RFSP-W,RFSP-I,RFCORE,
                   GFSP-N,GFSP-E,GFSP-S,GFSP-W,GFSP-I,GFCORE,)
               HEAT-S=ELECTRIC  Z-H-S=ELECTRIC
               S-O=COINCIDENT  S-R=1  ..
PLT-1     =P-A      S-N=(SYST-1)  ..

END ..
COMPUTE SYSTEMS ..

$-------------------------------------------------------
$ PLANT INPUT
$-------------------------------------------------------
INPUT PLANT
        INPUT-UNITS     METRIC
        OUTPUT-UNITS    METRIC ..     $—Specify metric units—

P-R  V=(PV-A)  S=(PS-A,PS-B,PS-C,PS-D,PS-E,PS-F,PS-G,BEPS) ..

$—PLANT BASIC PARAMETERS—
PARAMETER
    CHW-TEMP=6.7            $ Chilled water design temp drop (deg C)
    COP=1.2                 $ Chiller COP (kWe/TR)
    NOS-CHLR=6              $ Number of multiple identical chillers
    CHL-TYPE=HERM-REC-CHLR   $ Chiller type
    ..

$—PLANT DESCRIPTION—
$ 6 nos. air-cooled hermetic reciprocating chillers
$ automatically sized from peak load in system
$ SIZE= -290 in SI unit is equivalent to SIZE= -999 in English unit

PLT-1     =P-A  ..
CHIL-1    =P-E      TYPE=CHL-TYPE  SIZE= -290  I-N=NOS-CHLR
               M-N-A=NOS-CHLR  ..
PART-LOAD-RATIO     TYPE=CHL-TYPE  MIN-R=0.25  MAX-R=1  O-R=1
               E-I-R=COP TIMES .2844  ..

PLANT-PARAMETERS    CHILLER-CONTROL=DEMAND-ONLY   $—For chiller—
               HERM-REC-COND-TYPE=AIR
               HERM-REC-COND-PWR=0.03
               HERM-REC-UNL-RAT=0.25
               MIN-COND-AIR-T=18.33
               CHILL-WTR-T=CHW-TEMP
               CHILL-WTR-THROTTLE=1.39

               CCIRC-MOTOR-EFF=0.9          $—For chw pumps—
               CCIRC-IMPELLER-EFF=0.77
               CCIRC-HEAD=20
               CCIRC-DESIGN-T-DROP=5.56
               CCIRC-LOSS=0.01
               CCIRC-SIZE-OPT=SYSTEM-PEAK
               CCIRC-PUMP-TYPE=FIXED-SPEED
               CCIRC-MIN-PLR=0.5  ..

ENERGY-RESOURCE     R=ELECTRICITY  S-S-E=0.333  ..

END  ..
COMPUTE PLANT  ..
```